

Group Sequential Analysis Using the New SEQDESIGN and SEQTEST Procedures

Yang Yuan, SAS Institute Inc., Rockville, MD

ABSTRACT

In a fixed-sample clinical trial, data on all individuals are analyzed at the end of the study. In contrast, a group sequential trial provides for interim analyses before completion of the trial. Thus, a group sequential trial is useful for preventing unnecessary exposure of patients to an unsafe new drug or to a placebo treatment if a new drug shows significant improvement.

This paper reviews basic concepts of group sequential analysis and introduces two SAS/STAT[®] procedures: the SEQDESIGN and SEQTEST procedures. Both procedures are experimental in SAS[®] 9.2. The SEQDESIGN procedure creates group sequential designs by computing boundary values with a variety of methods, including the O'Brien-Fleming, Whitehead, and error spending methods; it also provides required sample sizes. The SEQTEST procedure compares the test statistic with the boundary values at each stage so that the trial can be stopped to reject or accept the hypothesis; it also computes parameter estimates, confidence limits, and p -values after the trial stops.

INTRODUCTION

A clinical trial is conducted according to a plan called a *protocol*. A protocol details the objectives of the trial, the data collection process, and the analysis. The protocol specifies the null hypothesis and an alternative hypothesis, a test statistic, the probability α of a Type I error, the probability β of a Type II error, the sample size needed to attain a specified power of $1 - \beta$ at an alternative reference, and critical values associated with the test statistic.

Most major trials have committees that periodically monitor safety and efficacy data during the trial and recommend that a trial be stopped for safety concerns such as an unacceptable toxicity level. In certain situations, the committee might recommend that a trial be stopped for efficacy. In contrast to a fixed-sample trial, a group sequential trial provides for interim analyses before the completion of the trial while maintaining the specified overall Type I and Type II error probabilities.


A group sequential trial is most useful in situations where it is important to monitor the trial to prevent unnecessary exposure of patients to an unsafe new drug, or alternatively to a placebo treatment if the new drug shows significant improvement. In most cases, if a group sequential trial stops early for safety concerns, fewer patients are exposed to the new treatment than in the fixed-sample trial. If a trial stops early for efficacy reasons, the new treatment is available sooner than it would be in a fixed-sample trial. Early stopping can also save time and resources.

A group sequential design provides detailed specifications for a group sequential trial. In addition to the usual specifications, it provides the total number of stages (the number of interim stages plus a final stage) and a stopping criterion to reject, accept, or either reject or accept the null hypothesis at each interim stage. It also provides critical values and the sample size at each stage for the trial.

At each interim stage, all the data collected up to that point are analyzed, and statistics such as a maximum likelihood test statistic and its associated standard error are computed. The test statistic is then compared with critical values generated from the sequential design, and the trial is stopped or continued. If a trial continues to the final stage, the null hypothesis is either rejected or accepted.

Figure 1 shows the boundaries (critical values) for a two-sided symmetric group sequential trial that stops early to reject the null hypothesis that the Treatment parameter is zero.

Figure 1 Boundaries for Two-Sided Group Sequential Test



The trial has four stages, which are indicated by the vertical lines labeled 1, 2, 3, and 4. With early stopping to reject the null hypothesis, the lower rejection boundary is constructed by connecting the lower critical values for the stages. Similarly, the upper rejection boundary is constructed by connecting the upper critical values for the stages. The horizontal axis indicates the sample size for the group sequential trial, and the vertical axis indicates the values of the test statistic on the standardized Z scale.

At each interim stage, if the test statistic falls into a rejection region (darker shaded area in Figure 1), the trial stops and the null hypothesis is rejected. Otherwise, the trial continues to the next stage. At the final stage (stage 4), the null hypothesis is rejected if Z falls into a rejection region. Otherwise, the null hypothesis is not rejected. In Figure 1, the test statistic does not fall into the rejection regions for stages 1 and 2, and so the trial continues to stage 3. At stage 3, the test statistic falls into the rejection region, and the null hypothesis is rejected.

A group sequential trial usually involves six steps:


1. You specify the statistical details of the design, including the null and alternative hypotheses, a test statistic for the hypothesis test, the Type I and II error probabilities, a stopping criterion, the total number of stages, and the relative information level at each stage.
2. You compute the boundary values for the trial based on the specifications in Step 1. You also compute the sample size required at each stage for the specified hypothesis test.
3. At each stage, you collect additional data with the required sample sizes. The data available at each stage include the data collected at the current stage in addition to the data collected at previous stages.
4. At each stage, you analyze the available data with a procedure such as the REG procedure, and you compute the test statistic.
5. At each stage, you compare the test statistic with the corresponding boundary values. You stop the trial to reject or accept the hypothesis, or you continue the trial to the next stage. If you continue the trial to the final stage, you either accept or reject the hypothesis.
6. After the trial stops, you compute parameter estimates, confidence limits for the parameter, and a p -value for the hypothesis test.

You use the SEQDESIGN procedure at Step 2 to compute the boundary values and required sample sizes for the trial. The boundary values are derived in such a way that the overall Type I and Type II error probability levels are maintained at the levels specified in the design. Steps 1 and 2 must be included as part of the protocol for the clinical trial.

You use the SEQTEST procedure at Step 5 to compare the test statistic with its boundary values. The boundary tables created in the SEQDESIGN procedure are structured for input to the SEQTEST procedure. You also use the SEQTEST procedure at Step 6 to compute parameter estimates, confidence limits, and p -values after the trial stops.

The flowchart in Figure 2 summarizes the steps in a typical group sequential trial and the relevant SAS procedures.

Figure 2 Group Sequential Trial



GROUP SEQUENTIAL DESIGNS

Suppose a group sequential trial is used to test the response difference between a treatment group and a placebo-controlled group. The responses for the treatment and control groups are assumed to be normally distributed with means μ_t and μ_c , respectively, and have a common variance σ^2 . The null hypothesis for the difference $\theta = \mu_t - \mu_c$ is $H_0 : \theta = 0$ with a two-sided alternative $H_1 : \theta \neq 0$.

Also suppose there are K stages in the trial and at each stage there are n additional individuals in each of the two groups. At stage k , the maximum likelihood estimate of θ is given by $\hat{\theta}_k = \bar{y}_{tk} - \bar{y}_{ck}$, the difference between two response averages. The estimate $\hat{\theta}_k$ has the variance

$$\text{Var}(\hat{\theta}_k) = \frac{2\sigma^2}{nk}$$

The Fisher information is the amount of information available about θ . For a maximum likelihood statistic $\hat{\theta}_k$, the information is the inverse of its variance

$$I_k = \frac{1}{\text{Var}(\hat{\theta}_k)} = \frac{nk}{2\sigma^2}$$

For a two-sided symmetric design with early stopping to reject the null hypothesis, the rejection boundary consists of two critical values at each stage. Denote the boundary values on the standardized Z statistic scale at stage k by $-a_k$ and a_k . Then the trial stops to reject H_0 for the lower alternative if $z_k \leq -a_k$; it stops to reject H_0 for the upper alternative if $z_k \geq a_k$; and it continues to the next stage if $-a_k < z_k < a_k$. If the trial continues to the final stage, H_0 is rejected if $z_K \leq -a_K$ or $z_K \geq a_K$; otherwise, H_0 is accepted. The boundary values are derived such that the overall Type I error level α and power $1 - \beta$ are maintained.

Armitage, McPherson, and Rowe (1969) showed that repeated significance tests at a fixed level on accumulating data increase the probability of obtaining a significant result under the null hypothesis. Pocock (1977) applied these repeated significance tests to group sequential trials with equally spaced information levels and derived a constant critical value on the standardized normal Z scale across all stages that maintains the Type I error probability level. In this situation, $a_k = c_{0\alpha}$, where $c_{0\alpha}$ is the derived constant. For example, with significance level 0.05 in a four-stage two-sided test, $a_k = 2.3613$, which is larger than the fixed-sample critical value 1.96.

O'Brien and Fleming (1979) proposed a sequential procedure with boundary values that decrease over the stages to make the early stopping less likely. This procedure has conservative stopping boundary values at very early stages, and boundary values at the final stage that are close to the fixed-sample design. In this situation, $a_k = \sqrt{K/k} c_{1\alpha}$, where $c_{1\alpha}$ is the derived constant. For example, with significance level 0.05 in a two-sided test, the critical values at the four stages are 4.0486, 2.8628, 2.3375, and 2.0243. Compared with the Pocock design, it is more difficult to reject the null hypothesis in early stages than in later stages.

Wang and Tsiatis (1987), Emerson and Fleming (1989) and Pampallona and Tsiatis (1994) generalized the Pocock and O'Brien-Fleming methods to the power family, which includes the Pocock and O'Brien-Fleming methods as special cases. The power family design uses boundary values

$$a_k = \left(\frac{k}{K}\right)^{-\rho} c_{2\alpha}$$

where the power parameter ρ controls the shape of the boundary and $c_{2\alpha}$ is the derived constant. For $\rho = 0$, the Pocock method is used, and for $\rho = 0.5$, the O'Brien-Fleming method is used.

Kittelson and Emerson (1999) extended the power family even further to the unified family, which includes the triangular method. The critical values on each boundary generated from the triangular method form a straight line for score statistics. The shape and location of each of the boundaries can be independently specified in the unified family methods.

Whitehead and Stratton (1983) and Whitehead (1997, 2001) developed triangular boundaries by adapting tests for continuous monitoring to discrete monitoring. Each boundary generated from the continuous monitoring forms a straight line for score statistics. With the discrete monitoring of group sequential designs, the critical values on each boundary are adjusted from the boundary line to have approximately the same α and β error probabilities. Note that the Whitehead triangular method is different from the unified family triangular method.

For a sequential design, you can derive α and β error probabilities at each stage from the boundaries. On the other hand, since a design is uniquely identified by the α and β errors spent at each stage, it can be derived by specifying the α and β error levels at each stage. Lan and DeMets (1983) introduced an approach based on an error spending function for specifying the error probabilities at each stage and then using these probabilities to derive the boundaries.

Numerous forms of the error spending function are available. Among them, Lan and DeMets (1983) proposed the Pocock-type and O'Brien-Fleming-type error spending functions, and Hwang, Shih, and DeCani (1990) proposed the gamma error spending function. In addition, the power error spending function (Jennison and Turnbull 2000, p. 148) uses

$$E(t; \rho) = \begin{cases} 1 & \text{if } t \geq 1 \\ t^\rho & \text{if } 0 < t < 1 \\ 0 & \text{otherwise} \end{cases}$$

That is, the cumulative α error spending at stage k is given by $\alpha E(\Pi_k; \rho)$, where $\Pi_k = I_k/I_K$, the information fraction at stage k . The boundaries created with $\rho = 1$ are similar to the boundaries from the Pocock method, and the boundaries created with $\rho = 3$ are similar to the boundaries from the O'Brien-Fleming method.

The following three groups of methods are available for computing boundaries with the SEQDESIGN procedure. You can specify a different method for each boundary, but all methods in a design must be from the same group.

- fixed boundary shape methods, which include the Haybittle-Peto method (Haybittle 1971; Peto et al. 1976) and

the unified family methods (which include the Pocock method, the O'Brien-Fleming method, the power family method, and the unified family triangular method)

- Whitehead triangular methods
- error spending methods

THE SEQDESIGN PROCEDURE

The SEQDESIGN procedure computes boundary values based on design specifications. It also computes the sample size required at each stage.

Statistical Assumptions for Group Sequential Designs

The SEQDESIGN procedure assumes that with a total number of stages K , the sequence of the standardized test statistics $\{Z_1, Z_2, \dots, Z_K\}$ has the canonical joint distribution with information levels $\{I_1, I_2, \dots, I_K\}$ for the parameter θ (Jennison and Turnbull 2000, p. 49):

- (Z_1, Z_2, \dots, Z_K) is multivariate normal
- $Z_k \sim N(\theta\sqrt{I_k}, 1)$, $k = 1, 2, \dots, K$
- $\text{Cov}(Z_{k_1}, Z_{k_2}) = \sqrt{(I_{k_1}/I_{k_2})}$, $1 \leq k_1 \leq k_2 \leq K$

If the test statistic is computed from data that are not from a normal distribution, such as a binomial distribution, then it is assumed that the test statistic is computed from a large sample such that the statistic has an approximately normal distribution.

Syntax: SEQDESIGN Procedure

The following statements are available in PROC SEQDESIGN:

```
PROC SEQDESIGN <options> ;  
  <label:> DESIGN options ;  
  SAMPLESIZE <MODEL=option> ;
```

Each DESIGN statement requests a new group sequential design, and you can use multiple DESIGN statements to create more than one design for comparison of features. The label identifies the design in the output tables and graphics. The SAMPLESIZE statement computes the required sample sizes for the design specified in each DESIGN statement.

The following are some of the options you can use in the PROC SEQDESIGN statement.

ALTREF= θ_1

specifies the alternative reference—that is, the value of θ at which the power is computed.

BOUNDARYSCALE=MLE | SCORE | STDZ | PVALUE

specifies the scale for the statistic displayed in boundary tables and boundary plots. The keywords MLE, SCORE, STDZ, and PVALUE correspond to the maximum likelihood estimate scale, the score statistic scale, the standardized normal Z scale, and the p -value scale, respectively. The default is BOUNDARYSCALE=STDZ.

PLOTS <(ONLY)> <= plot-request >

PLOTS <(ONLY)> <= (plot-request < ... plot-request >) >

specifies options that control plots. The global plot option ONLY suppresses the default plots and displays only plots specifically requested. The option ALL displays all appropriate plots, and the option NONE suppresses all plots. The options ASN, BOUNDARY, COMBINEDBOUNDARY, ERRSPEND, and POWER respectively display a plot of average sample numbers (expected sample sizes) under various hypothetical references, a plot of boundaries with acceptance and rejection regions for each design, a plot of boundaries for all designs, a plot of error spending for all designs, and a plot of power curves under various hypothetical references. The default is PLOTS=BOUNDARY.

DESIGN Statement: SEQDESIGN Procedure

< label: > **DESIGN** < options > ;

The DESIGN statement requests a group sequential design. [Table 1](#) lists some of the available options.

Table 1 Design Statement Options

Option	Description
Design Parameters	
ALPHA=	specifies the Type I error probability level α
ALT=	specifies the type of alternative hypothesis
BETA=	specifies the Type II error probability level β
INFO=	specifies the information levels
NSTAGES=	specifies the number of stages
STOP=	specifies the condition for early stopping
Boundary Methods	
METHOD=	specifies the methods for boundary values

The ALPHA= and BETA= options specify the Type I and Type II error levels, respectively. The NSTAGES= option specifies the number of stages, which include the interim stages and the final stage. The remaining options are as follows:

ALT=LOWER | UPPER | TWOSIDED

specifies the type of alternative hypothesis. The keywords LOWER, UPPER, and TWOSIDED correspond to the alternatives $\theta < 0$, $\theta > 0$, and $\theta \neq 0$, respectively. The default is ALT=TWOSIDED.

INFO=EQUAL

INFO=CUM(numbers)

specifies relative information levels for all stages. The INFO=EQUAL option specifies equally spaced information levels, and the INFO=CUM option specifies relative cumulative information levels. The default is INFO=EQUAL.

STOP=ACCEPT | REJECT | BOTH

specifies the condition for early stopping. The keywords ACCEPT, REJECT, and BOTH correspond to early stopping only to accept, only to reject, and either to accept or reject the null hypothesis H_0 , respectively. The default is STOP=REJECT.

METHOD=method

METHOD(boundary) = method

specifies the methods for deriving the boundaries. You use METHOD= to specify the same method for all boundaries in the design. Alternatively, you use METHOD(boundary)= to specify different methods from the same group for the various boundaries.

[Table 2](#) lists available methods for deriving the boundaries. The methods are displayed in three distinct groups.

Table 2 Available METHOD= Options

Option	Description
Fixed Boundary Shape Methods	
OBF	specifies the O'Brien-Fleming method
PETO	specifies the Haybittle-Peto method
POC	specifies the Pocock method
POW	specifies a power family method
TRI	specifies a unified family triangular method
UNI	specifies a unified family method
Error Spending Methods	
ERRFUNCGAMMA	specifies a gamma error spending function
ERRFUNCOBF	specifies the O'Brien-Fleming-type error spending function
ERRFUNCPOC	specifies the Pocock-type error spending function

Table 2 *continued*

Option	Description
ERRFUNCPOW	specifies a power error spending function
ERRSPEND	specifies the relative error spending at each stage
Whitehead Method	
WHITEHEAD	specifies the Whitehead triangular method

SAMPLESIZE Statement: SEQDESIGN Procedure

SAMPLESIZE < MODEL= option > ;

The SAMPLESIZE statement computes the required sample sizes for various models. Table 3 lists some of the options available in the SAMPLESIZE statement.

Table 3 SAMPLESIZE Statement Options

Option	Description
Fixed-Sample Size	
INPUTNOBS ($N = n$)	specifies the sample size for fixed-sample design
INPUTNEVENTS ($D = d$)	specifies the number of events for fixed-sample design
One-Sample Models	
ONESAMPLEMEAN	specifies the one-sample Z test for mean
ONESAMPLEFREQ	specifies the one-sample test for binomial proportion
Two-Sample Models	
TWOSAMPLEMEAN	specifies the two-sample Z test for mean difference
TWOSAMPLEFREQ	specifies the two-sample test for binomial proportions
TWOSAMPLESURVIVAL	specifies the log-rank test for two survival distributions
Regression Models	
REG	specifies the test for a regression parameter
LOGISTIC	specifies the test for a logistic regression parameter
PHREG	specifies the test for a proportional hazards regression parameter

You use the MODEL= option to explicitly specify the sample size or number of events for a fixed-sample study, or to specify a statistical model that is used to compute the required sample size. The MODEL=INPUTNOBS option specifies the input sample size for a fixed-sample study of nonsurvival data, and the MODEL=INPUTNEVENTS option specifies the number of events for a fixed-sample study of survival data. The corresponding sample size or number of events for a group sequential trial is computed by multiplying the derived information ratio (which is the ratio between the design information level and its corresponding fixed-sample information level) by the input sample size or number of events. The other MODEL= options specify statistical models that are used to compute the required sample size. The default is MODEL=TWOSAMPLEMEAN.

EXAMPLE 1: CREATING A TWO-SIDED O'BRIEN-FLEMING DESIGN

This example illustrates a two-sided O'Brien-Fleming design used to stop a trial early for ethical concerns about possible harm or for unexpectedly strong efficacy of the new drug.

Suppose that a clinical trial is conducted to test the efficacy of a new cholesterol-lowering drug. The primary focus is low-density lipoprotein (LDL), the so-called bad cholesterol, which is a risk factor for coronary heart disease. LDL is measured in *mg/dL*, milligrams per deciliter of blood. The trial consists of two groups of equally allocated patients with elevated LDL levels: an experimental group given the new drug and a placebo-controlled group.

Suppose that changes in LDL level after the treatment are normally distributed with means μ_e and μ_c for individuals in the experimental and control groups, respectively, and that the changes have a common variance σ^2 . Then the null hypothesis of no effect for the new drug is $H_0 : \theta = 0$, where $\theta = \mu_e - \mu_c$.

For a fixed-sample design with sample size n in each group, the MLE for θ is $\hat{\theta} = \hat{\mu}_e - \hat{\mu}_c$, where $\hat{\mu}_e$ and $\hat{\mu}_c$ are the sample means of the changes in LDL level in the experimental and control groups, respectively. The statistic $\hat{\theta}$ has a normal distribution

$$\hat{\theta} \sim N\left(\theta, \frac{2\sigma^2}{n}\right)$$

Thus, under the null hypothesis $H_0 : \theta = 0$, the standardized test statistic is

$$Z = \frac{\hat{\theta}}{\sqrt{\frac{2\sigma^2}{n}}} \sim N(0, 1)$$

The Z statistic can be used to test the null hypothesis H_0 . If the variance σ^2 is unknown, the sample variance can be substituted for σ^2 if n is large enough that the Z statistic has an approximately standard normal distribution.

With a two-sided Type I error probability $\alpha = 0.05$, the critical values for Z are given by $\Phi^{-1}(\alpha/2) = -1.96$ and $\Phi^{-1}(1 - \alpha/2) = 1.96$, where Φ is the cumulative standard normal distribution function. At the end of study, if $Z \geq 1.96$, the null hypothesis is rejected for harmful drug effect, and if $Z \leq -1.96$, the null hypothesis is rejected for efficacy of the new drug. Otherwise, the null hypothesis is not rejected.

Also suppose that the alternative reference $\theta = -10$ is a clinically meaningful difference that the trial should detect with a high probability (power). Furthermore, suppose that a reasonable estimate of the standard deviation is $\hat{\sigma} = 20$. The following statements invoke the SEQDESIGN procedure and request a four-stage O'Brien-Fleming design:

```
ods graphics on;
proc seqdesign altref=-10
    boundaryscale=stdz
    plots=boundary(hscale=samplesize)
    ;
    TwoSidedOBrienFleming: design method=obf
        alt=twosided stop=reject
        nstages=4
        alpha=0.05 beta=0.10
    ;
    samplesize model=twosamplemean(stddev=20);
ods output Boundary=bound_ldl;
run;
ods graphics off;
```

The ALTREF= option specifies the alternative reference, and the BOUNDARYSCALE=STDZ option specifies the standardized Z scale for values in the boundary information table and the boundary plot. Note that for a symmetric two-sided design, the ALTREF=-10 option implies a lower alternative reference of -10 and an upper alternative reference of 10 . With the ODS GRAPHICS ON statement, the PLOTS=BOUNDARY option displays the boundary plot, which is shown in [Figure 6](#).

The label "TwoSidedOBrienFleming" identifies the design in the output. The STOP=REJECT option specifies early stopping in the interim stages only for rejecting the null hypothesis. That is, at each interim stage, the trial is stopped to reject the null hypothesis or is continued to the next stage.

The ALT=TWOSIDED option specifies a two-sided alternative hypothesis. There are two boundaries: an upper α boundary that consists of upper rejection critical values and a lower α boundary that consists of lower rejection critical values. With METHOD=OBF, the O'Brien-Fleming method is used to compute the two boundaries for the design.

The NSTAGES=4 option specifies the total number of stages, including three interim stages and a final stage. In the SEQDESIGN procedure, the null hypothesis is $H_0 : \theta = 0$. The ALPHA= option specifies the Type I error probability $\alpha = 0.05$. The BETA= option specifies the Type II error probability $\beta = 0.10$ corresponding to a power of $1 - \beta = 0.90$ at the alternative reference $H_1 : \theta = -10$.

The MODEL=TWOSAMPLEMEAN option computes required sample sizes for a two-sample test for mean difference. The ODS OUTPUT statement with the BOUNDARY=BOUND_LDL option creates an output data set named BOUND_LDL which contains the resulting boundary information.

By default, the SEQDESIGN procedure derives boundary values with equally spaced information levels for all stages—that is, the same information increment between successive stages. The “Design Information,” “Method Information,” and “Boundary Information” tables are displayed by default, as shown in Figure 3, Figure 4, and Figure 5, respectively.

The “Design Information” table in Figure 3 displays design specifications and derived statistics of the maximum information and the average sample numbers under the null and alternative hypotheses.

Figure 3 O'Brien-Fleming Design Information

The SEQDESIGN Procedure	
Design: TwoSidedOBrienFleming	
Design Information	
Statistic Distribution	Normal
Boundary Scale	Standardized Z
Alternative Hypothesis	Two-Sided
Early Stop	Reject Null
Method	O'Brien-Fleming
Boundary Key	Both
Alternative Reference	-10
Number of Stages	4
Alpha	0.05
Beta	0.1
Power	0.9
Max Information (Percent of Fixed Sample)	102.2163
Max Information	0.107403
Null Ref ASN (Percent of Fixed Sample)	101.5728
Alt. Ref ASN (Percent of Fixed Sample)	76.7397

Max Information is the information level at the final stage. Max Information (Percent of Fixed Sample) is the maximum information for the sequential design expressed as a percentage of the information for the corresponding fixed-sample design. Here, the information needed for the group sequential trial is 2.22% more than that of the corresponding fixed-sample design if the trial does not stop at any interim stage.

Null Ref ASN (Percent of Fixed Sample) is the average sample size required under the null hypothesis for the group sequential design expressed as a percentage of the sample size for the corresponding fixed-sample design. Here, the average sample size for the group sequential trial is 1.57% greater than the corresponding fixed-sample size under the null hypothesis.

Similarly, Alt. Ref ASN (Percent of Fixed Sample) is the average sample size required under the alternative hypothesis for the group sequential design expressed as a percentage of the sample size for the corresponding fixed-sample design. Here, the average sample size for the group sequential trial is 76.74% of the corresponding fixed-sample size if the alternative hypothesis is true.

In this example, the O'Brien-Fleming design requires only a slight increase in sample size if the trial proceeds to the final stage. On the other hand, if the alternative hypothesis is correct, this design provides a substantial saving in sample size.

The “Method Information” table in Figure 4 displays the computed Type I error probability α , Type II error probability β , and the derived drift parameters for the lower and upper α boundaries.

Figure 4 Method Information

Method Information						
Boundary	Method	Alpha	Beta	-----Unified Family-----		
				Rho	Tau	C
Upper Alpha	O'Brien-Fleming	0.02500	0.10000	0.5	0	2.02429
Lower Alpha	O'Brien-Fleming	0.02500	0.10000	0.5	0	2.02429
Method Information						
	Boundary	Alternative Reference	Drift			
	Upper Alpha	10	3.277238			
	Lower Alpha	-10	-3.27724			

The drift parameter is the standardized alternative reference at the final stage $\theta_1 \sqrt{I_K}$, where θ_1 is the alternative reference specified with the ALTREF= option and I_K is the maximum information. Drift parameters are derived even if the alternative reference is not specified.

The O'Brien-Fleming method belongs to the unified family of designs, which is parameterized by two parameters, ρ and τ , as implemented in the SEQDESIGN procedure. The "Method Information" table displays the values $\rho = 0.5$ and $\tau = 0$, which are the parameters for the O'Brien-Fleming method. The table also displays the derived parameter $C_\alpha = 2.02429$, which is used in the construction of symmetric lower and upper α boundaries.

The "Boundary Information" table in Figure 5 displays the lower and upper boundary values at each stage. The table also displays the information level, including the proportion, actual level, and corresponding sample size (N) at each stage.

Figure 5 Boundary Information

Boundary Information (Standardized Z Scale)					
Null Reference = 0					
Stage	-----Information Level-----			-----Alternative-----	
	Proportion	Actual	N	Lower	Upper
1	0.2500	0.026851	42.96116	-1.63862	1.63862
2	0.5000	0.053701	85.92233	-2.31736	2.31736
3	0.7500	0.080552	128.8835	-2.83817	2.83817
4	1.0000	0.107403	171.8447	-3.27724	3.27724


Boundary Information (Standardized Z Scale)			
Null Reference = 0			
Stage	-----Boundary Values-----		
	---Lower---	---Upper---	
	Alpha	Alpha	
1	-4.04859	4.04859	
2	-2.86278	2.86278	
3	-2.33745	2.33745	
4	-2.02429	2.02429	

The table also displays the lower and upper alternative references. With BOUNDARYSCALE=STDZ, the alternative references and boundary values are displayed with the standardized Z statistic scale. The alternative reference on this scale at stage k is given by $\theta_1 \sqrt{I_k}$, where θ_1 is the alternative reference and I_k is the information available at stage k , $k = 1, 2, 3, 4$. By default, equally spaced information levels for all stages are used to derive boundary values.

In this example, a standardized Z statistic is computed by standardizing the estimate of the effect in LDL level, and a negative Z test statistic indicates a beneficial effect. Consequently, at each interim stage, if the standardized Z test statistic is less than or equal to the corresponding lower α boundary value, the hypothesis H_0 is rejected for efficacy. If the test statistic is greater than or equal to the corresponding upper α boundary value, the hypothesis H_0 is rejected for harmful effect. Otherwise, the process continues to the next stage. At the final stage (stage 4), the hypothesis H_0 is rejected for efficacy if the Z statistic is less than or equal to the corresponding lower α boundary value -2.02429 , and the hypothesis H_0 is rejected for harmful effect if the Z statistic is greater than or equal to the corresponding upper α boundary value 2.02429 . Otherwise, the H_0 is not rejected.

When you specify the ODS GRAPHICS ON statement and the PLOTS=BOUNDARY(HSCALE=SAMPLESIZE) option, a boundary plot is displayed, as shown in Figure 6. This plot displays the values in the “Boundary Information” table. The horizontal axis indicates the sample sizes for the stages, and the stages are indicated by vertical lines with accompanying stage numbers.

Figure 6 Boundary Plot



In the boundary plot, if a test statistic is in the rejection region (darker shaded area), the trial stops and the null hypothesis is rejected. If the test statistic is in the acceptance region (lightly shaded area), the trial stops and the hypothesis is not rejected. The symbol “o” identifies the fixed-sample critical values of -1.96 and 1.96 , and the accompanying vertical line indicates the required sample size for the fixed-sample design. Note that the boundary values ± 2.02429 at the final stage are close to the fixed-sample critical values ± 1.96 .

When you provide the SAMPLESIZE statement, the maximum information is used to compute the required sample sizes. The MODEL=TWOSAMPLEMEAN(STDDEV=20) option specifies the test for this computation.

The “Sample Size Summary” table in Figure 7 displays the parameters for the sample size computation and the resulting maximum and expected sample sizes at the final stage.

Figure 7 Sample Size Summary

Sample Size Summary	
Test	Two-Sample Means
Mean Difference	10
Std. Dev.	20
Max Sample Size	171.8447
Expected Sample Size (Null Ref)	170.7627
Expected Sample Size (Alt. Ref)	129.0137

The “Sample Sizes (N)” table in [Figure 8](#) displays the required sample sizes at each stage, as both fractional and integer numbers. By default, the sample sizes for the two groups are equal for the test.

Figure 8 Derived Sample Sizes

Sample Sizes (N) Two-Sample Z Test for Mean Difference				
-----Fractional N-----				
Stage	N	N(Grp 1)	N(Grp 2)	Information
1	42.96	21.48	21.48	0.0269
2	85.92	42.96	42.96	0.0537
3	128.88	64.44	64.44	0.0806
4	171.84	85.92	85.92	0.1074

Sample Sizes (N) Two-Sample Z Test for Mean Difference				
-----Ceiling N-----				
Stage	N	N(Grp 1)	N(Grp 2)	Information
1	44	22	22	0.0275
2	86	43	43	0.0538
3	130	65	65	0.0812
4	172	86	86	0.1075

In practice, integer sample sizes are used, and so the resulting information levels increase slightly.

THE SEQTEST PROCEDURE

The SEQTEST procedure performs interim analyses by comparing test statistics with corresponding boundary values obtained with the SEQDESIGN procedure. If the information levels for the test statistics do not match the information levels obtained with the SEQDESIGN procedure, the SEQTEST procedure modifies the original boundary values to adjust for actual (observed) information levels.

Syntax: SEQTEST Procedure

The following statement is used with the SEQTEST procedure:

```
PROC SEQTEST <options> ;
```

[Table 4](#) lists some of the options in the PROC SEQTEST statement. The BOUNDARY= option provides an input data set with boundary information for the design, which is usually created from the “Boundary Information” table in the SEQDESIGN procedure. The DATA= option names a SAS data set that contains the test statistics and information levels. The PARMS= option names a SAS data set that contains the parameter estimates and their associated standard errors, which are used to derive the information levels.

Table 4 PROC SEQTEST Options

Option	Description
Input Data Sets	
BOUNDARY=	specifies a data set with boundary information
DATA=	specifies a data set with test statistics and information levels
PARMS=	specifies a data set with parameter estimates and standard errors
Boundaries and Sample Space Ordering	
BOUNDARYADJ=	specifies the method for boundary adjustment for information levels
BOUNDARYKEY=	specifies the boundary key to maintain Type I and II error probability levels
BOUNDARYSCALE=	specifies the boundary scale
NSTAGES=	specifies the number of stages
ORDER=	specifies the ordering for the computation of p -values and confidence limits

You use the following options to derive boundary values and to specify the sample space ordering.

BOUNDARYADJ=method

BOUNDARYADJ(boundary)=method

specifies the method for boundary adjustments if the information level at a stage does not match the level in the BOUNDARY= data set. You use BOUNDARYADJ= to specify the same method for all boundaries in the design. Alternatively, you use BOUNDARYADJ(boundary)= to specify different methods for the various boundaries.

If the actual information levels in a group sequential trial do not match the levels in the original design, the cumulative error spending values at these and successive stages are derived from the cumulative error spending values in the original design. These derived cumulative error spending values are then used to create the boundary values for the trial. Table 5 lists available methods for boundary adjustments. The default is BOUNDARYADJ=ERRLINE.

Table 5 Available BOUNDARYADJ= Options

Option	Description
Boundary Adjustment Methods	
NONE	specifies that the cumulative error spending at each interim stage not be changed
ERRLINE	specifies the linear interpolation method for the adjustment (Kittelson and Emerson 1999)
ERRFUNCGAMMA	specifies the gamma function method for the adjustment
ERRFUNCOBF	specifies the O'Brien-Fleming-type error spending function for the adjustment
ERRFUNCPOC	specifies the Pocock-type error spending function for the adjustment
ERRFUNCPOW	specifies the power error spending function for the adjustment

BOUNDARYKEY=ALPHA | BETA | BOTH

specifies the boundary key to be maintained in the boundary adjustments. The BOUNDARYKEY=ALPHA option maintains the Type I α level and derives the Type II error probability, and the BOUNDARYKEY=BETA option maintains the Type II β level and derives the Type I error probability. The BOUNDARYKEY=BOTH option maintains both α and β levels simultaneously by deriving a new maximum information level. The default is BOUNDARYKEY=ALPHA.

BOUNDARYSCALE=MLE | SCORE | STDZ | PVALUE

specifies the boundary scale to be displayed in the tables and plots. The keywords MLE, SCORE, STDZ, and PVALUE correspond to the maximum likelihood estimate scale, the score statistic scale, the standardized normal Z scale, and the p -value scale, respectively. The default is BOUNDARYSCALE=STDZ.

NSTAGES=number

specifies the number of stages for the clinical trial. This number might or might not be the same as the number in the BOUNDARY= data set. The default is the total number of stages in the BOUNDARY= data set.

ORDER=LR | MLE | STAGEWISE

specifies the ordering of the sample space (k, z) , where k is the stage number and z is the observed standardized Z statistic, for test statistics that result in the stopping of a trial (that is, for all statistics in the rejection region and in the acceptance region). The ordering is used to derive the p -value for the observed statistics (k, z) that results in the stopping of a trial and to create the unbiased median estimate and confidence limits from the statistics (k, z) . The keywords LR, MLE, and STAGEWISE specify the LR (likelihood ratio) ordering, the MLE ordering, and the stage-wise ordering, respectively. The default is ORDER=STAGEWISE.

The stage-wise ordering (Fairbanks and Madsen 1982; Tsiatis, Rosner, and Mehta 1984) orders statistics in the same rejection or acceptance region by their stage numbers first. For example, in a one-sided design with an upper alternative, for two statistics at different stages in the rejection region, the statistic in the early stage is more extreme; for two statistics at the same stage, the statistic with a larger Z value is more extreme. The stage-wise ordering does not require the boundary information at future stages, but it is not applicable for statistics in the acceptance region of a two-sided design.

The LR ordering (Rosner and Tsiatis 1988; Chang 1989) orders the sample space by the standardized Z statistic z , and the MLE ordering (Emerson and Fleming 1990) orders the sample space by the maximum likelihood estimates $z/\sqrt{I_k}$. The LR and MLE orderings are applicable for all designs, but they require the boundary information at future stages to be available at the stopping stage.

EXAMPLE 2: PERFORMING GROUP SEQUENTIAL TESTS

This example is a continuation of Example 1. Here, a standardized Z statistic is computed at each stage by standardizing the parameter estimate of the effect in LDL level. Note that in a typical trial, the actual information levels do not match the information levels specified in the design. The SEQTEST procedure modifies the boundary values stored in the BOUND_LDL data set to adjust for these new information levels.

Suppose that there are 22 individuals available in each group at stage 1. Figure 9 lists the first six observations in the data set ldl_1.

Figure 9 Partial Data at Stage 1

First 6 Observations in the Trial Data		
Obs	Trt	Ldl
1	0	33.33
2	1	-14.89
3	0	15.30
4	1	4.71
5	0	26.89
6	1	-48.74

The variable Trt is an indicator variable with value 1 for individuals in the treatment group and value 0 for individuals in the placebo-controlled group. The variable Ldl is the effect on the LDL level for each individual.

The following statements use the REG procedure to estimate the mean treatment difference and standard error at the first stage:

```
proc reg data=ldl_1;
  model Ldl=Trt;
  ods output ParameterEstimates=parms_1;
run;
```

The following statements create the data set for the mean treatment difference and its associated standard error. Figure 10 shows the results.

```
data ldlparms_1;
  set parms_1;
  if Variable='Trt';
  keep _Scale_ _Stage_ Variable Estimate StdErr;
  _Scale_='MLE';
  _Stage_= 1;
run;
proc print data=ldlparms_1;
  title 'Stage 1 Test Statistics';
run;
```

Figure 10 Stage 1 Test Statistics

Stage 1 Test Statistics					
Obs	Variable	Estimate	StdErr	_Scale_	_Stage_
1	Trt	-2.52591	5.68572	MLE	1

The following statements invoke the SEQTEST procedure to test for early stopping at the first stage:

```
ods graphics on;
proc seqtest boundary=bound_ldl
  parms(testvar=Trt)= ldlparms_1
  order=stagewise
  ;
ods output test=tldl_1;
run;
ods graphics off;
```

The BOUNDARY= option specifies the input data set that contains the boundary information generated by the SEQDESIGN procedure. The PARM= option specifies the input data set that contains the test statistic and standard error for each stage. The information level for the test statistic is computed from the standard error. By default, the total number of stages for the trial is the same as the number derived from the BOUNDARY= data set. The ORDER=STAGEWISE option specifies the sample space ordering.

The ODS OUTPUT statement with the TEST=TLDL_1 option creates an output data set named TLDL_1 which contains the resulting boundary information for group sequential tests at stage 1 and future stages.

The “Design Information” table in Figure 11 displays design specifications and derived statistics of maximum information and average sample numbers under the null and alternative hypotheses.

Figure 11 Design Information

The SEQTEST Procedure	
Design Information	
BOUNDARY Data Set	WORK.BOUND_LDL
Data Set	WORK.LDLPARMS_1
Statistic Distribution	Normal
Boundary Scale	Standardized Z
Alternative Hypothesis	Two-Sided
Early Stop	Reject Null
Number of Stages	4
Alpha	0.05
Beta	0.10074
Power	0.89926
Max Information (Percent of Fixed Sample)	102.4815
Max Information	0.10740291
Null Ref ASN (Percent of Fixed Sample)	101.7765
Alt. Ref ASN (Percent of Fixed Sample)	75.4928

With the default BOUNDARYKEY=ALPHA option, the boundary values are modified for actual information levels to maintain α . The maximum information remains the same as in the BOUNDARY= data set, but the information levels at subsequent interim stages are adjusted proportionally. The derived Type II error probability β and power $1 - \beta$ are slightly different with actual information levels.

The “Test Information” table in Figure 12 displays boundary values by using the default standardized Z scale. Information levels at future interim stages are derived proportionally from the corresponding levels in the BOUNDARY= data set.

Figure 12 Group Sequential Test Information


Test Information (Standardized Z Scale)						
Null Reference = 0						
Stage	--Information Level--		-----Alternative-----		---Boundary Values---	
	Proportion	Actual	-----Reference-----	Upper	---Lower---	---Upper---
			Lower		Alpha	Alpha
1	0.2880	0.030934	-1.75879	1.75879	-3.39532	3.39532
2	0.5253	0.056423	-2.37536	2.37536	-2.77374	2.77374
3	0.7627	0.081913	-2.86205	2.86205	-2.32412	2.32412
4	1.0000	0.107403	-3.27724	3.27724	-2.03147	2.03147

Test Information (Standardized Z Scale)		
Null Reference = 0		
Stage	-----Test-----	
	Estimate	Action
1	-0.44426	Continue
2	.	
3	.	
4	.	

At stage 1, the standardized Z statistic -0.44426 lies between the lower and upper α boundary values, and so the trial continues to the next stage. Since the observed information level at stage 1, $I_1 = 0.0309$, is only slightly larger than the target information level in the design, the trial can continue to the next stage without an adjustment of the sample size according to the study plan. If an observed information level differs substantially from its target level, then the sample sizes should be adjusted in the subsequent stages.

When you specify the ODS GRAPHICS ON statement, a test plot with rejection and acceptance regions is displayed by default, as shown in Figure 13. This plot displays the boundary values in the “Test Information” table. The stages are indicated by vertical lines with accompanying stage numbers. The horizontal axis indicates the information levels for the stages. The test statistic lies in the continuation region between the lower and upper α boundaries.

Figure 13 Sequential Test Plot



Suppose that at stage 2, the data set LDLPARMS_2 contains the mean treatment difference -8.37628 and an associated standard error 4.24405. The following statements invoke the SEQTEST procedure to test for early stopping at the second stage:

```
ods graphics on;
proc seqtest boundary=tldl_1
    parms( testvar=Trt)= ldlparms_2
    order=stagewise
;
ods output test=tldl_2;
run;
ods graphics off;
```

Since the standardized test statistic -1.97365 lies between its corresponding lower and upper α boundary values (not shown), the trial continues to the next stage. The ODS OUTPUT statement with the TEST=TLDL_2 option creates an output data set named TLDL_2 which contains the resulting boundary information for group sequential tests at stage 2 and future stages.

The following statements use the REG procedure to estimate the mean treatment difference at the third stage:

```
proc reg data=ldl_3;
    model Ldl=Trt;
ods output ParameterEstimates=parms_3;
run;
```


The following statements create and display the data set (shown in [Figure 14](#)) with the test statistics for stage 3:

```

data ldlparms_3;
  set parms_3;
  if Variable='Trt';
  keep _Scale_ _Stage_ Variable Estimate StdErr;
  _Scale_='MLE';
  _Stage_= 3;
run;
data ldlparms_3;
  set ldlparms_2 ldlparms_3;
run;
proc print data=ldlparms_3;
  title 'Stage 3 Test Statistics';
run;

```

Figure 14 Stage 3 Test statistics

Stage 3 Test Statistics					
Obs	Variable	Estimate	StdErr	_Scale_	_Stage_
1	Trt	-2.52591	5.68572	MLE	1
2	Trt	-8.37628	4.24405	MLE	2
3	Trt	-9.21369	3.42149	MLE	3

The following statements invoke the SEQTEST procedure to test for early stopping at the third stage:

```

ods graphics on;
proc seqtest boundary=tld1_2
  parms( testvar=Trt)= ldlparms_3
  order=stagewise
  ;
ods output test=tld1_3;
run;
ods graphics off;

```

The “Test Information” table is shown in [Figure 15](#).

Figure 15 Group Sequential Test Information


The SEQTEST Procedure						
Test Information (Standardized Z Scale)						
Null Reference = 0						
Stage	--Information Level--		-----Alternative-----		----Boundary Values----	
	Proportion	Actual	-----Reference-----	Lower	Upper	Lower
			Lower	Upper	Alpha	Alpha
1	0.2880	0.030934	-1.75879	1.75879	-3.39532	3.39532
2	0.5169	0.055519	-2.35624	2.35624	-2.78456	2.78456
3	0.7953	0.085422	-2.92271	2.92271	-2.25480	2.25480
4	1.0000	0.107403	-3.27724	3.27724	-2.04573	2.04573

Test Information (Standardized Z Scale)		
Null Reference = 0		
Stage	-----Test-----	
	Estimate	Action
1	-0.44426	Continue
2	-1.97365	Continue
3	-2.69289	Reject Null
4	.	.

Since the test statistic -2.69289 is less than the corresponding lower α boundary -2.25479 , the test stops at stage 3 to reject the null hypothesis for the lower alternative. That is, the test demonstrates significant beneficial effect for the new drug.

The “Test Plot” for the first three stages is shown in Figure 16.

Figure 16 Sequential Test Plot



The “Parameter Estimates” table in Figure 17 displays the stopping stage, parameter estimate, p -value under the null hypothesis $H_0 : \theta = 0$, unbiased median estimate, and confidence limits. The ORDER=STAGewise option specifies the stage-wise ordering of the sample space used to compute the p -value, unbiased median estimate, and confidence limits. As expected, the p -value 0.0108 is significant at the two-sided α level 0.05, and the confidence interval does not contain the value zero.

Figure 17 Parameter Estimates

Parameter Estimates Stage-Wise Ordering				
Parameter	Stopping Stage	MLE	p-Value for H0:Parm=0	Median Estimate
Trt	3	-9.213692	0.0108	-9.022900
Parameter Estimates Stage-Wise Ordering				
Parameter	95% Confidence Limits			
Trt	-15.79845	-2.13138		

CONCLUSION

This paper reviews basic concepts of group sequential analysis and introduces two SAS/STAT procedures: the SEQDESIGN and SEQTEST procedures. The SEQDESIGN procedure creates group sequential designs by computing boundary values with a variety of methods; it also provides required sample sizes. The SEQTEST procedure compares the test statistic with the boundary values at each stage so that the trial can be stopped to reject or accept the hypothesis; it also computes parameter estimates, confidence limits, and p -values after the trial stops.

ACKNOWLEDGMENT

The author is grateful to Bob Rodriguez and Anne Jones of the Advanced Analytics Division at SAS Institute for their valuable assistance in the preparation of this manuscript.

REFERENCES

- Armitage, P., McPherson, C. K., and Rowe, B. C. (1969), "Repeated Significance Test on Accumulating Data," *Journal of the Royal Statistical Society, Series A*, 132, 235–244.
- Chang, M. N. (1989), "Confidence Intervals for a Normal Mean Following a Group Sequential Test," *Biometrics*, 45, 247–254.
- Emerson, S. S. and Fleming, T. R. (1989), "Symmetric Group Sequential Designs," *Biometrics*, 45, 905–923.
- Emerson, S. S. and Fleming, T. R. (1990), "Parameter Estimation Following Group Sequential Hypothesis Testing," *Biometrika*, 77, 875–892.
- Fairbanks, K. and Madsen, R. (1982), "P Values for Tests Using a Repeated Significance Test Design," *Biometrika*, 69, 69–74.
- Haybittle, J. L. (1971), "Repeated Assessment of Results in Clinical Trials of Cancer Treatment," *British Journal of Radiology*, 44, 793–797.
- Hwang, I. K., Shih, W. J., and DeCani, J. S. (1990), "Group Sequential Designs Using a Family of Type I Error Probability Spending Functions," *Statistics in Medicine*, 9, 1439–1445.
- Jennison, C. and Turnbull, B. W. (2000), *Group Sequential Methods with Applications to Clinical Trials*, New York: Chapman & Hall.
- Kittelson, J. M. and Emerson, S. S. (1999), "A Unifying Family of Group Sequential Test Designs," *Biometrics*, 55, 874–882.
- Lan, K. K. G. and DeMets, D. L. (1983), "Discrete Sequential Boundaries for Clinical Trials," *Biometrika*, 70, 659–663.
- O'Brien, P. C. and Fleming, T. R. (1979), "A Multiple Testing Procedure for Clinical Trials," *Biometrics*, 35, 549–556.
- Pampallona, S. and Tsiatis, A. A. (1994), "Group Sequential Designs for One-Sided and Two-Sided Hypothesis Testing with Provision for Early Stopping in Favor of the Null Hypothesis," *Journal of Statistical Planning and Inference*, 42, 19–35.
- Peto, R., Pike, M. C., Armitage, P., Breslow, N. E., Cox, D. R., Howard, S. V., Mantel, N., McPherson, K., Peto, J., and Smith, P. G. (1976), "Design and Analysis of Randomized Clinical Trials Requiring Prolonged Observation of Each Patient: I. Introduction and Design," *British Journal of Cancer*, 34, 585–612.
- Pocock, S. J. (1977), "Group Sequential Methods in the Design and Analysis of Clinical Trials," *Biometrika*, 64, 191–199.
- Rosner, G. L. and Tsiatis, A. A. (1988), "Exact Confidence Intervals Following a Group Sequential Trial: A Comparison of Methods," *Biometrika*, 75, 723–729.
- Tsiatis, A. A., Rosner G. L., and Mehta C. R. (1984), "Exact Confidence Intervals Following a Group Sequential Test," *Biometrics*, 40, 797–803.
- Wang, S. K. and Tsiatis, A. A. (1987), "Approximately Optimal One-Parameter Boundaries for Group Sequential Trials," *Biometrics*, 43, 193–200.
- Whitehead, J. (1997), *The Design and Analysis of Sequential Clinical Trials*, Revised Second Edition, Chichester: John Wiley & Sons.
- Whitehead, J. (2001), "Use of the Triangular Test in Sequential Clinical Trials," *Handbook of Statistics in Clinical Oncology*, ed. J. Crowley, New York: Marcel Dekker, 211–228.
- Whitehead, J. and Stratton, I. (1983), "Group Sequential Clinical Trials with Triangular Continuation Regions," *Biometrics*, 39, 227–236.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author:

Yang Yuan
SAS Institute Inc.
111 Rockville Pike, Suite 1000
Rockville, MD 20852
E-mail: Yang.Yuan@sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are registered trademarks or trademarks of their respective companies.